<?xml version="1.0" encoding="gb2312"?>
<xsl:stylesheet version="1.0" xmlns:xsl="http://www.w3.org/1999/XSL/Transform" xmlns:tk="http://www.qdocuments.net" xmlns:msxsl="urn:schemas-microsoft-com:xslt" exclude-result-prefixes="tk msxsl">
  <xsl:import href="DetailItems_Trans.xslt"/>
  <xsl:import href="PublicItems.xslt"/>
  <xsl:template name="tk:IsTwoColumn">
    <xsl:param name="IsTwoColumn"/>
    <xsl:choose>
      <xsl:when test="$IsTwoColumn">
        <xsl:attribute name="width">35%</xsl:attribute>
      </xsl:when>
      <xsl:otherwise>
        <xsl:attribute name="colspan">3</xsl:attribute>
      </xsl:otherwise>
    </xsl:choose>
  </xsl:template>
  <xsl:template name="tk:DetailValue">
    <xsl:param name="HRefType"/>
    <xsl:param name="HRefAddress"/>
    <xsl:param name="Value"/>
    <xsl:param name="Target"/>
    <b>
      <xsl:choose>
        <xsl:when test="$HRefType='HREF'">
          <a href="{$HRefAddress}">
            <xsl:if test="$Target!=''">
              <xsl:attribute name="target"><xsl:value-of select="$Target"/></xsl:attribute>
            </xsl:if>
            <xsl:value-of select="$Value"/>
          </a>
        </xsl:when>
        <xsl:when test="$HRefType='HTTP'">
          <a target="_blank">
            <xsl:attribute name="href"><xsl:if test="not(starts-with(translate($HRefAddress,'htp','HTP'), 'HTTP'))">http://</xsl:if><xsl:value-of select="$HRefAddress"/></xsl:attribute>
            <xsl:value-of select="$Value"/>
          </a>
        </xsl:when>
        <xsl:when test="$HRefType='EMAIL'">
          <a href="mailto:{$Value}">
            <xsl:value-of select="$Value"/>
          </a>
        </xsl:when>
        <xsl:otherwise>
          <xsl:call-template name="tk:ProcessText">
            <xsl:with-param name="Text" select="$Value"/>
          </xsl:call-template>
        </xsl:otherwise>
      </xsl:choose>
    </b>
  </xsl:template>
  <xsl:template name="tk:Hidden">
    <xsl:param name="Field"/>
    <xsl:param name="Value"/>
    <input type="hidden" name="{$Field/tk:FieldName}" id="{$Field/tk:FieldName}" value="{$Value}"/>
  </xsl:template>
  <xsl:template name="tk:Detail">
    <xsl:param name="Field"/>
    <xsl:param name="Value"/>
    <xsl:call-template name="tk:Caption">
      <xsl:with-param name="Title" select="$Field/tk:DisplayName"/>
    </xsl:call-template>
    <td class="td-detail">
      <xsl:call-template name="tk:IsTwoColumn">
        <xsl:with-param name="IsTwoColumn" select="$Field/tk:Display/@IsTwoColumn"/>
      </xsl:call-template>
      <xsl:choose>
        <xsl:when test="$Field/tk:Extension/@CheckValue!=''">
          <xsl:if test="$Value=$Field/tk:Extension/@CheckValue">
            <b>
              <IMG src="../images/check.gif" border="0" width="16" height="16"/>
            </b>
          </xsl:if>
        </xsl:when>
        <xsl:otherwise>
          <xsl:variable name="Link" select="$Field/tk:Display/Link"/>
          <xsl:variable name="RefValue">
            <xsl:for-each select="$Value">
              <xsl:value-of select="../*[local-name()=$Link/@ID]"/>
            </xsl:for-each>
          </xsl:variable>
          <xsl:variable name="HRefType">
            <xsl:call-template name="tk:HRefType">
              <xsl:with-param name="Link" select="$Link"/>
              <xsl:with-param name="RefValue" select="$RefValue"/>
            </xsl:call-template>
          </xsl:variable>
          <xsl:variable name="HRefAddress">
            <xsl:call-template name="tk:HRefAddress">
              <xsl:with-param name="HRefType" select="$HRefType"/>
              <xsl:with-param name="Link" select="$Link"/>
              <xsl:with-param name="RefValue" select="$RefValue"/>
              <xsl:with-param name="Value" select="$Value"/>
            </xsl:call-template>
          </xsl:variable>
          <xsl:variable name="DisplayValue">
            <xsl:call-template name="tk:DisplayValue">
              <xsl:with-param name="Field" select="$Field"/>
              <xsl:with-param name="Value" select="$Value"/>
            </xsl:call-template>
          </xsl:variable>
          <xsl:call-template name="tk:DetailValue">
            <xsl:with-param name="HRefType" select="$HRefType"/>
            <xsl:with-param name="HRefAddress" select="$HRefAddress"/>
            <xsl:with-param name="Value" select="$DisplayValue"/>
            <xsl:with-param name="Target" select="$Link/@Target"/>
          </xsl:call-template>
        </xsl:otherwise>
      </xsl:choose>
    </td>
  </xsl:template>
  <xsl:template name="tk:DetailUpload">
    <xsl:param name="Field"/>
    <xsl:param name="DataRecord"/>
    <xsl:call-template name="tk:Caption">
      <xsl:with-param name="Title" select="$Field/tk:DisplayName"/>
    </xsl:call-template>
    <td class="td-detail">
      <xsl:call-template name="tk:IsTwoColumn">
        <xsl:with-param name="IsTwoColumn" select="$Field/tk:Display/@IsTwoColumn"/>
      </xsl:call-template>
      <b>
        <xsl:call-template name="tk:DisplayUpload">
          <xsl:with-param name="Field" select="$Field"/>
          <xsl:with-param name="DataRecord" select="$DataRecord"/>
          <xsl:with-param name="FieldList" select="ancestor::tk:Table/*/tk:Field"/>
        </xsl:call-template>
      </b>
    </td>
  </xsl:template>
</xsl:stylesheet>
